
Introduction 
 
Group insurance is becoming a greater focus area for many insurers operating in India.  Greater 
focus has lead to greater competition among players, and players constantly look for ways to 
make their product offering more attractive to potential group buyers.  One such method for doing 
this is to offer the scheme a share in any profits that arise over a period of time.   
 
While such offerings may be attractive to potential group schemes, the actual cost (or increase in 
base premium required) for offering such a scheme is often not well understood by insurers and 
potential clients alike. 
 
This paper proposes a simple mathematical framework for calculating the  the cost of profit a 
stochastic basis.  A description of the process is given, as well as the Visual Basic source code 
for a simple model. 
 

Profit share variables 
 
In calculating profit shares, profit is usually defined as: 
 
Profit   =  Premiums Earned, less  
    Claims Incurred, less  
    Expenses, less  
    Losses carried forward from a previous period. 
 
The use of “claims incurred” rather than “claims paid” would mean that some allowance in the 
formula would need to be made for claims “incurred but not reported” or “IBNR”.  However, if the 
profit share is calculated at a date after which final claim liability has been agreed,  IBNR could be 
ignored. 
 
In this formula, expenses are usually expressed as a percentage of premiums earned, but might 
also include some elements based on some fixed amount, or some amount relative to sums at 
risk. 
 
Whether previous years’ losses are included in the profit formula is a matter for some discussion.  
It is usually argued that such a structure is impractical, since any scheme that suffers a loss in 
any one year would be inclined to move their coverage to a new insurer with a similar profit share 
structure.  In doing so, future profits in the new scheme would not be reduced by any such loss.  
Therefore, losses from previous years are often ignored. 
 



Profit share = Profit Refund Percentage  *  Profit 
 
The profit refund percentage might be constant, or might increase based on ratio of Profit to 
Premiums Earned.  Sometimes, a layered approach to the profit share percentage is used. 
 



Profit share pricing 
 
The principle in pricing for profit share correctly is that the additional cost of profit share should 
ensure that the insurance company has the same absolute expectation of technical result as 
would have been the case if there was no profit share 
 

Volatility of profit result 
 
Volatility is created by the possibility that experience  might emerge differently from how we may 
have expected it to.  The less sure we are that experience will emerge as expected, the greater 
the expected volatility of result.   
 
Given the need to price for the same expected technical result in a profit sharing environment as 
a non participating one, the volatility of profit that is expected to emerge is of paramount 
importance.  The greater the volatility of expected scheme result, the greater will be the cost of 
the profit share.  As such, the expected volatility of result needs to be modeled in order to 
calculate the cost of profit share.  We therefore need to look a the likely distributions of outcomes 
to assess the impact on volatility. 
 

Claims number distributions 
 
In a group of n identical lives with identical sums assured, with each life having the probability p 
of dying in any one year, the expected claims result should statistically follow a binomial 
distribution. 
 
In such a case, the number of deaths that we would expect to see would be n*p. 
 
The probability that k deaths occur in the year would be: 

The probability that k or more deaths occur in a year would be: 

As such, the variance that would be expected from such distribution would be n * p * (1-p) 
 
Given that probability of death p is so small, there is very little difference between the mean and 
variance the distributed number of deaths.  Therefore a common alternatively used distribution of 
claims outcomes that is often a lot easier to model than the binomial distribution is the Poisson 
distribution (whose mean and variance are the same).  This leads to a simpler unbiased 
probability function for the number of deaths that occur in a year: 

The model attached to this paper assumes that claim numbers follow a Poisson distribution. 



Sums assured distributions 
 
Assuming a constant probability of claim, if all lives in a group had the same sum assured, there 
would be no difference in the impact on profitability of one death versus another.  However, 
usually group life schemes have variation in sum assured, with more higher salaried staff having 
higher sums assured (and higher ages).   
 
Various distributions can be considered in modeling salary distributions across a portfolio of lives, 
but the most commonly used one is a lognormal distribution. For modeling purposes, lognormal 
parameters are set based on the mean and variance of the actual sums assured of a portfolio.  
Hence, the logarithmic ln(salary) is distributed normally with mean as follows: 

An alternative that is sometimes used is to set up the model to use the exact sums assured of the 
group in question.  However, this can lead to significant increases in simulation time. 
 
However, of interest is not the distribution of the possible salaries for any one life that dies, but 
the distribution of the sum of the possible salaries of all lives that die in any one year.  This can 
be modeled exactly but looking at repeated lognormal distributions, or by using a method such as 
the Box-Müller method (which is the approach adopted in the model attached to this paper). 
 
An issue often ignored in modeling profitability of an group insurance scheme is the correlation 
between the claim probability and the sum assured.  In reality, older lives tend to have higher 
sums assured.  Higher mortality of older lives means that should a death occur, it is more likely 
that it was an older person, and hence the sum assured is more likely to be higher than average. 
 
Without complicating the model further by modeling age distributions and resultant variation in 
mortality, the above issue can be taken into account by setting the best estimate claims 
assumption used in the model to be the sum assured weighted average expected mortality 
decrement. 
 



Factors affecting claim cost volatility 
 
In pricing for profit share, it is not so much the actual variance or standard deviation of the claim 
cost distribution that affects result, it is the relationship between the standard deviation and the 
mean.  Looking at just the claim numbers distribution, we need to look at the volatility ratio: 

Given that claim cost volatility is driven by this value, any factor that affects either the variables n 
or p have an impact on the claim cost volatility. 
 
Size of scheme 
 
As n increases the volatility ratio reduces.  Hence the larger the scheme, the lower the volatility. 
 
Age distribution of the scheme 
 
Given that mortality increases with age, as age (and p) increases, the volatility ratio decreases.  
Hence an older scheme on average might have a lower cost of profit share. 
 
Occupational classification of the scheme 
 
As for age, heavier mortality of industrial schemes make cost of profit share cheaper than for high 
tech industries and other class 1 occupation schemes. 
 
Distribution of sums assured 
 
Over and above the claim numbers distribution, the volatility of overall claim result is increased by 
increased variance in sums assured. 
 

Other factors to consider in setting up a model 
 
Confidence in best estimate mortality 
 
Assuming a set level of best estimate mortality implies that modeling of expected profitability only 
looks at the systematic risk of fluctuations around that best estimate, and does not look at the risk 
that the best estimate assumption of itself is incorrect.  If there is some level of uncertainty when 
setting best estimate mortality assumptions, then any profits resulting from a conservative best 
estimate assumption is shared with the scheme, where as losses resulting from a potentially 
lenient or aggressive assumption are not shared.   
 
The most common approach for dealing with this risk is to use a slightly lighter best estimate 
mortality assumption in modeling the cost of profit share than was used in calculating the cost of 
the base premium. How much lighter would depend on the confidence in the originally placed in 
the assumption. 
 



Interest 
 
Premium is usually received at the beginning of a scheme year, with claims being paid out mid 
year on average, and profit share returned at the end of the year.  When equating expected 
margin in premium between profit sharing structures and non-participating structures, the timing 
of cashflows also needs to be taken into account. 
 

The profit simulation 
 
A model is created to calculate the expected result of a group scheme based on the above 
distributions.  Each simulation would caluclate the profit margin arising from a scheme as: 
 
Profit = Premiums Received – Expenses – Sum of Claims (Number of claims) 
 
Premiums Received and Expenses would be fixed for each simulation. 
 
Sum of Claims (Number of claims) would be the simulated amount based on the above discussed 
distributions.  Firstly, a “Number of Claims” is simulated based on the poisson distribution 
described, and from that number, a “Sum of Claims” is calculated  based on the sum of 
lognormals.   
 

The profit distribution  and profit share calculation process 
 
Repeated simulations of this profit result gives a distribution of profits around an expected mean.  
For each of these scenarios, a calculation is made of the profit share refund that would apply.  
This profit share refund amount is then added and averaged over all the simulations and then 
compared with the original premium amount to determine what profit share loading should have 
applied in order to fund that profit refund. This profit share loading is then applied to the original 
premium, and the whole simulation repeated on this new premium to calculate the new profit 
share refund proportion.  This process is repeated again until the increases in premium required 
in successive iterations reaches a targeted minimum.  
 
The above short description can be explained in more detail and broken down pictorially as 
shown below: 

1.  As the number of simulations increases, the 
distribution of possible profits/results tends 
towards normal, with the mean being the 
expected level of margin over the expected 
claim cost and expenses. 

2.  The area under the curve represents the 
overall profitability of the scheme.  Where a 
profit on the scheme is expected, an amount of 
profit refund needs to be returned to the 
scheme.  This is represented by the mustard 
area shown on the graph. 

3.  From this profit amount, the expense 
allowance from the profit share formula must be 
removed.  The shaded area now represents the 
cost of the profit share for the given simulation. 

4.  If the cost of profit share is added to the 
premium and re-simulated, a different set of 
possible profitability outcomes is generated. 

5.  This in turn creates a different (larger) cost 
of profit share, since given the larger premium, 
a larger proportion of that premium will be 
returned as profit. 
 
 
 
 
 
 
6.  However, a large portion of this cost of profit 
share has already been charged for and so it is 
only the remaining amount that must be added, 
and the process re-iterated. 

7.  This process continues until a the 
consecutive differences in resultant additional 
premium required become insignificant.  

  
  

  
  
  
  
  
 



An Excel model interface 
 
A simple Excel model can be created to perform the simulations required.  Such a model would 
use Visual basic to run the simulation  based on inputs from the Excel moded.  A sample inputs 
screen might look as follows: 
 
Simple Group Profit share calculator

Portfolio Parameters Basic Premium parameters

Reinsured Portfol io Size 2 ,000 Best Est imate Claim Rate 2 .000 per mille

Average Sum Insured 200 ,000 Per Mi l le expense 0 . 2 0        

Std Dev Sum Insured 200 ,000 Net Premium Loading 5 .00%
Gross Premium Loading 7 .00%

Profit Share Formula  a(eP-C) Final Gross Premium 2.992      

Percentage Profit (a) 50% Simulation parameters
Percentage Expenses (e) 90%

Number of Simulations 4 0 , 0 0 0    

Calcu lated WP Loading 21 .0% Interest 5 . 0 %

Ca lcu la te  load ing

 

Conclusion 
 
Calculating the exact cost of a profit share option in pricing a group scheme can be done on a 
technical basis.  A sample set of visual basic code that could be used for calculating profit share 
based on the above description is included in this paper as an appendix. 
 
An Excel model for use in pricing is available from the author on request. 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
Note:  The views, models and assumptions expressed in this paper are those of the author alone 
and do not represent those of his employer.



Sample Source Code 
 
'Simple Group Profit Share Pricing Model  
'Written by Walter de Oude 
'Prepared for presentation at 7th Global Actuarial Seminar - India Feb 2005 
 
Option Explicit 
 
Public PortSize, FinalPremRate, ClaimRate As Double 
Public NetPremMargins, PerMillePremLoad, GrossPremMargins As Double 
Public ProfitShareLoading As Double 
Public NonProfitPremRate As Double 
 
Public AveSumInsured As Double 
Public StDevSumInsured As Double 
Public LogMu As Double 
Public LogSigma As Double 
 
Public PSPercProf As Double 
Public PSPercExp As Double 
 
Public i, j, Simulations As Long 
Public Interest As Double 
 
Public ExpectedNumberOfClaims As Double 
Public NumberOfClaims() As Integer 
Public ClaimsCost() As Double 
Public Premium As Double 
 
Public ProfitShare() As Double 
Public Profit() As Double 
 
Public AverageNumberOfClaims() As Double 
Public AverageClaimsCost() As Double 
 
Public TotalPVExpectedProfit As Double 
Public AveragePVProfit As Double 
 
Public Const Pi = 3.14159265358979 
 
Sub CalculatePremium() 
 
NonProfitPremRate = (ClaimRate * (1 + NetPremMargins) + PerMillePremLoad) / _ 
                (1 - GrossPremMargins) 
FinalPremRate = NonProfitPremRate * (1 + ProfitShareLoading) 
TotalPVExpectedProfit = PortSize * (NonProfitPrem Rate * (1 - GrossPremMargins) + _ 
 FinalPremRate * GrossPremMargins - ClaimRate * (1 + Interest) ^ (-0.5)) * _ 
 AveSumInsured 
 
End Sub 
 

Sub Participation() 
    InitialiseVariables 
    Rnd (-1) 
' Rnd(-1) is done to make sure that the same result is generated 
' for the same set of parameter inputs every time 
    CalculatePremium  
    AveragePVProfit = 1 
    While Abs(TotalPVExpectedProfit / AveragePVProfit - 1) > 0.003 
' We repeat the process until the result comes close to an equilibrium  
        SimulateClaimNumbers 
' to get a number of claims 
        SimulateClaimsCost 
' for the number of claims, what is the resultant cost 
        CalculateProfitShare 
        CalculatePremium 
    Wend 
End Sub 
 
Sub InitialiseVariables() 
' This procedure takes inputs required as tabulated in the Excel spreadsheet 
    PortSize = Range("PortSize").Value  'The size of the portfolio 
    ClaimRate = Range("ClaimRate").Value / 1000   'The rate of mortality decrement 
    AveSumInsured = Range("AveSumInsured").Value   ' Average sum insured of the portfolio 
    StDevSumInsured = Range("StDevSumInsured").Value   
    NetPremMargins = Range("NetPremMargins").Value   'Loading applicable to claim rate         
    PerMillePremLoad = Range("PerMillePremLoad").Value / 1000 'e.g. Capital charges       
    GrossPremMargins = Range("GrossPremMargins").Value 'e.g. commissions 
    PSPercProf = Range("PSPercProf").Value 'The percentage of profit refunded 
    PSPercExp = Range("PSPercExp").Value  'The percentage of premium used in profit calc  
    Simulations = Range("Simulations").Value ' Number of simulations 
    Interest = Range("Interest").Value 
    ProfitShareLoading = 0 
    LogSigma = (Log((StDevSumInsured / AveSumInsured) ^ 2 + 1)) ^ 0.5 
    LogMu = Log(AveSumInsured) - 0.5 * LogSigma ^ 2 
End Sub 
 
Sub SimulateClaimNumbers()  
    ExpectedNumberOfClaims = PortSize * ClaimRate 
    ReDim NumberOfClaims(Simulations) As Integer 
    For j = 1 To Simulations 
        NumberOfClaims(j) = Poisson(ExpectedNumberOfClaims) 
    Next j  
End Sub 
 
Sub SimulateClaimsCost()  
    ReDim ClaimsCost(Simulations) As Double 
    For j = 1 To Simulations 
        ClaimsCost(j) = SumLogNormal(Nu mberOfClaims(j)) 
    Next j  
End Sub 
 
Sub CalculateProfitShare() 
    ReDim ProfitShare(Simulations) As Double 
    ReDim Profit(Simulations) As Double 



    Dim ProfitAccount As Double 
    AveragePVProfit = 0 
    Premium = PortSize * FinalPremRate * AveSumInsured 
        For j = 1 To Simulations 
            ProfitAccount = PSPercExp * Premium - ClaimsCost(j) 
            If ProfitAccount > 0 Then 
                ProfitShare(j) = PSPercProf * ProfitAccount 
            Else 
                ProfitShare(j) = 0 
            End If 
            Profit(j) = Premium - ClaimsCost(j) * (1 + Interest) ^ (-0.5) - _ 
                        ProfitShare(j) * (1 + Interest) ^ (-1) 
            AveragePVProfit = AveragePVProfit + Profit(j) 
        Next j 
    AveragePVProfit = AveragePVProfit / Simulations 
    Premium = Premium + TotalPVExpectedProfit - AveragePVProfit 
    ProfitShareLoading = Premium / (NonProfitPremRate * PortSize * AveSumInsured) - 1 
    Range("PSLoading").Value = ProfitShareLoading 
End Sub 
 
Function Poisson(L As Double) As Integer 
' Generate Random Number from a Poisson Distribution 
    Dim PoiTarg As Double 
    Dim PoiRes As Double 
    Dim PoiCount As Integer 
 
    PoiTarg = Exp(-L) 
    PoiCount = 0 
    PoiRes = Rnd 
    Do While PoiRes > PoiTarg 
        PoiRes = PoiRes * Rnd 
        PoiCount = PoiCount + 1 
    Loop 
    Poisson = PoiCount 
End Function 
 
Function SumLogNormal(N As Integer) As Double 
' Generate Sum of N LogNormal(LogMu, LogSigma) Random Variables 
' Box-Müller Method, see 
' Box, G. E. P. and Muller, M. E. "A Note on the Generation of 
' Random Normal Deviates." Ann. Math. Stat. 29, 610-611, 1958. 
     
    Dim X1, X2 As Double 
    Dim R1, R2 As Double 
    Dim LNRes As Double 
        
    LNRes = 0 
    If N = 0 Then 
        LNRes = 0 
    ' Even Number of Claims 
    ElseIf N Mod 2 = 0 Then 
        For i = 1 To N / 2 
            R1 = Rnd 
            If R1 = 0 Then 

                R1 = 1E-16 
            End If 
            R2 = Rnd 
            X1 = (-2 * Log(R1)) ^ 0.5 * Cos(2 * Pi * R2) 
            X1 = Exp(X1 * LogSigma + LogMu) 
            X2 = (-2 * Log(R1)) ^ 0.5 * Sin(2 * Pi * R2) 
            X2 = Exp(X2 * LogSigma + LogMu) 
            LNRes = LNRes + X1 + X2 
        Next i 
    ' Odd Number of Claims 
    Else  
        For i = 1 To (N - 1) / 2 
            R1 = Rnd 
            If R1 = 0 Then 
                R1 = 1E-16 
            End If 
            R2 = Rnd 
            X1 = (-2 * Log(R1)) ^ 0.5 * Cos(2 * Pi * R2) 
            X1 = Exp(X1 * LogSigma + LogMu) 
            X2 = (-2 * Log(R1)) ^ 0.5 * Sin(2 * Pi * R2) 
            X2 = Exp(X2 * LogSigma + LogMu) 
            LNRes = LNRes + X1 + X2 
        Next i 
        R1 = Rnd 
        If R1 = 0 Then 
            R1 = 1E-16 
        End If 
        R2 = Rnd 
        X1 = (-2 * Log(R1)) ^ 0.5 * Cos(2 * Pi * R2) 
        X1 = Exp(X1 * LogSigma + LogMu) 
        LNRes = LNRes + X1 
    End If 
    SumLogNormal = LNRes 
End Function 
 
 


